Knowledge Representation and Semantic Technologies

Exercises on Datalog and ASP

Riccardo Rosati

Corso di Laurea Magistrale in Ingegneria Informatica
Sapienza Università di Roma
2020/2021

Exercise 1

Given the following positive Datalog program P:

```
r(X,Y) :- s(X,Y).
r(X,Y) :- r(X,Z), s(Z,Y).
t(X) :- r(X,X).
q(Y) :- t(X), r(X,Y).
s(a,b).
s(b,c).
s(c,a).
```

1) compute the minimal model of P;
2) tell if atom q(a) is entailed by P.

Exercise 1 - Solution

To compute the minimal model, we use the semi-naive evaluation method. We first define the program P' with Δ-relations:

Δ'r(X,Y) :- Δr(X,Z), s(Z,Y). [rule R1]
Δ't(X) :- Δr(X,X). [rule R2]
Δ'q(Y) :- Δt(X), r(X,Y). [rule R3]
Δ'q(Y) :- t(X), Δr(X,Y). [rule R4]
s(a,b). s(b,c). s(c,a).

Exercise 1 - Solution

We then execute the iterative computation of the intensional predicates r, t, q through semi-naive evaluation on P':

Initialization:

$I = \{ s(a,b), s(b,c), s(c,a) \}$,

$I' = T_P(I) = I \cup \{ r(a,b), r(b,c), r(c,a) \}$ (using 1st rule of P),

$\Delta'I = \{ \Delta'r(a,b), \Delta'r(b,c), \Delta'r(c,a) \}$

1st execution of the repeat-until loop:

$I = I \cup \{ r(a,b), r(b,c), r(c,a) \}$,

$\Delta I = \{ \Delta r(a,b), \Delta r(b,c), \Delta r(c,a) \}$

$\Delta'I = T_{\Delta P}(I) = \{ \Delta'r(a,c), \Delta'r(b,a), \Delta'r(c,b) \}$ (using rule R1)

Exercise 1 - Solution

2nd execution of the repeat-until loop:

$I = I \cup \{ r(a,c), r(b,a), r(c,b) \}$,

$\Delta I = \{ \Delta r(a,c), \Delta r(b,a), \Delta r(c,b) \}$

$\Delta' I = T_{\Delta P}(I) = \{ \Delta' r(a,a), \Delta' r(b,b), \Delta' r(c,c) \}$ (using rule R1)

3rd execution of the repeat-until loop:

$I = I \cup \{ r(a,a), r(b,b), r(c,c) \}$,

$\Delta I = \{ \Delta r(a,a), \Delta r(b,b), \Delta r(c,c) \}$

$\Delta' I = T_{\Delta P}(I) = \{ \Delta' t(a), \Delta' t(b), \Delta' t(c) \}$ (using rule R2)

Exercise 1 - Solution

4th execution of the repeat-until loop:

$I = I \cup \{ t(a), t(b), t(c) \}$,

$\Delta I = \{ \Delta t(a), \Delta t(b), \Delta t(c) \}$

$\Delta' I = T_{\Delta P}(I) = \{ \Delta' q(a), \Delta' q(b), \Delta' q(c) \}$ (using rule R3)

5th execution of the repeat-until loop:

$I = I \cup \{ q(a), q(b), q(c) \}$,

$\Delta I = \{ \Delta q(a), \Delta q(b), \Delta q(c) \}$

$\Delta' I = T_{\Delta P}(I) = \{ \}$

Exercise 1 - Solution

The minimal model of P is thus the following:

MM(P) = { s(a,b), s(b,c), s(c,a), r(a,b), r(b,c), r(c,a), r(a,c),
r(b,a), r(c,b), r(a,a), r(b,b), r(c,c), t(a), t(b), t(c),
q(a), q(b), q(c) }

Exercise 1 - Solution

Finally, since atom q(a) belongs to the minimal model of P, it is entailed by P.

Exercise 2

Given the following positive Datalog program with constraints P':

```
r(X,Y) :- s(X,Y).
r(X,Y) :- r(X,Z), s(Z,Y).
t(X) :- r(X,X).
q(Y) :- t(X), r(X,Y).
:- t(X), q(X).
s(a,b). s(b,c). s(c,a).
```

compute the minimal model of P'.

Exercise 2 - Solution

We notice that the program P' is the same as the positive program of Exercise 1, plus the constraint :- t(X), q(X). Namely, P'= P $\cup$ { :- t(X), q(X) }

So, to answer the question we only have to check whether the minimal model of P satisfies such a constraint.

Exercise 2 - Solution

The minimal model M of P (see Exercise 1) is:

MM(P) = { s(a,b), s(b,c), s(c,a), r(a,b), r(b,c), r(c,a), r(a,c),
r(b,a), r(c,b), r(a,a), r(b,b), r(c,c), t(a), t(b), t(c),
q(a), q(b), q(c) }

M does not satisfy the constraint :- t(X), q(X).
(e.g., both t(a) and q(a) belong to M).

So, we conclude that there exists no (minimal) model for P.